
色々なループ
二重ループ
　Forループの中にForループを書くことができます。2重ループと呼びます。

＜例題1＞
　九九の表を作りなさい。
Sub kuku_table()
    For i = 1 To 9
        For j = 1 To 9
            Cells(i, j) = i * j
        Next j
    Next i
End Sub
　内側のループの変数はj、外側のループの変数はiです。異なる変数名を使う必要があります。

＜課題1＞

のように、0から99までの数値でセルA1からJ10を埋めるプログラムを作成しなさい。

＜課題2＞
　電気回路において2つの抵抗R1とR2を並列に接続したとき、合成抵抗RはR=R1*R2/(R1+R2)で求められます。R1<R2という条件で、どちらも1～100の整数のとき、合成抵抗が整数となる組み合わせを全て求め、表を作成するプログラムを作りなさい。
　表はA列にR1の値、B列にR2の値、C列に合成抵抗の値を書き込むものとします。
　変数aが整数か否かは
　　　If a = Int(a) Then
あるいは
　　　If a = Round(a,0) Then
で確認することができます。
無限ループ
　Forループはあらかじめ決まった回数を繰り返すループでした。ここでは、無限にループを繰り返す方法とそのループから抜ける方法を学習します。

＜例題2＞
　1, 2, 4, 8, …… という数列をA列に作成しなさい。ただし、10000未満の数についての表を作りなさい。

学ぶ命令：　Do While, Loop, Exit Do
Sub mugen_loop()

    row = 1
    num = 1

    Do While True
        Range("A" & row) = num
        row = row + 1
        num = num * 2
        If num >= 10000 Then
            Exit Do
        End If
    Loop

End Sub
　Do While True ～ Loopの部分が無限ループです。プログラムはこの部分を無限に繰り返します。ループから抜けるにはExit Do命令を使います。Exit Doを実行すると、Loopの次の文から実行します。ここではループを1回まわるごとに変数rowの値を1ずつ増加させ、numの値を2倍します。numが10000以上になったらループを抜けます。
　Forループから抜ける必要が生じることは、通常はないと思いますが、次のようにします。
    For i = 1 To 10
        Debug.Print i
        If i = 6 Then
            Exit For
        End If
    Next i
＜課題3＞
　セルB1に1, B2に1を入れなさい。B3以下にフィボナッチ数列を作るプログラムを作成しなさい。ただし、10000以下の数値のみでよい。
ElseIf
　整数aの内容によって以下の処理をする場合を考えます。
10 <= a のとき処理A
5 <= a < 10のとき処理B
0 <= a < 5のとき処理C
それ以外のとき処理D
　If文の中にIf文を書くことができます。これまでに習った命令を用いて上記の処理を書くと以下のようになります。
    If a >= 10 Then
        処理A
    Else
        If a >= 5 Then
            処理B
        Else
            If a >= 0 Then
                処理C
            Else
                処理D
            End If
        End If
    End If
　これは少々見づらいです。ElseIf命令を使って以下のように書き直せます。
    If a >= 10 Then
        処理A
    ElseIf a >= 5 Then
        処理B
    ElseIf a >= 0 Then
        処理C
    Else
        処理D
    End If
　IfのあとにElseIfを連ねる書き方は
条件が成立した場合はThen以下を実行してこのIf文は終わり
成立しない場合は次々に下の条件を判定してゆく
という処理を意味しています。慣れるまでは分かりにくく感じるかも知れませんが、プログラムの定番パターンの一つなので、慣れましょう。

＜課題4＞
　VBA2.xlsmのシート「成績処理」を使います。

1. 中間テストと期末テストの平均点、最高点、最低点を求めるプログラムを作成しなさい。
2. 上記の結果を確認するため、関数Average, Max, Minを使って平均点、最高点、最低点を求める数式をB15～B17, D15～D17に記入しなさい。
3. 中間と期末の最終得点を求める数式をC列とE列に記入しなさい。欠席時の点数は0点です。関数IFを使います。
4. テスト点を求める数式をF列に記入しなさい。テスト点は　(中間テスト＋期末テスト) ÷ 2　です。
5. レポート点を記入するマクロを作成しなさい。Aのとき10点、Bのとき8点、Cのとき6点です。
6. 総合得点は「テスト点×0.8 ＋ レポート点×2 」です。四捨五入して整数にします。総合得点を記入する数式を書きなさい。四捨五入はROUND関数を使います。
7. 成績は85～100点のときA、70～84点のときB、60～69点のときC、60未満はDです。また、テストを1回でも休んだ人はDとなります。成績を書き込むマクロを作成しなさい。
シートの操作
　シートを切り替えるには、次のようにします。
    Sheets("シート名").Select      名前で指定
    Sheets(2).Select               番号で指定　左端が1
　シートの個数は以下のように得ます。
    n = Worksheets.Count
　今選択しているシートの名前を取得するには次のようにします。
    sheet_name = ActiveSheet.name
　シート上の領域を選択するには以下のように操作します。
    Cells.Select                全領域を選択
    Range("B2:C4").Select      一部分を選択
　選択した領域をクリアするには以下のようにします。
    Selection.ClearContents    セルを空にするが、罫線・塗りつぶし・列幅は
                                  残る。
    Selection.Clear             列幅のみ残る
　列幅もデフォルトに戻したいときは、以下のようにします。
    Cells.Select
    Selection.Delete
　あるシートを別のシートにコピーするには、次のようにします。
    Sheets("コピー元").Select
    Cells.Select
    Selection.Copy
    Sheets("コピー先").Select
    Cells.Select
    ActiveSheet.Paste
　シートAを操作しているときに、シートBのセルC2に "abc" を入れるときは、以下のように書きます。
    Sheets("B").Range("C2") = "abc"
　A列において、データが入っている最終行を取り出すには、以下のように書きます。
Range("A65536").End(xlUp).Select
last_row = ActiveCell.Row
　B列の最終行は、Range("B65536") とします。

＜例題4＞
　シート「成績処理2」「レポート」を使います。シート「レポート」にレポート提出者のリストがあります。レポート提出者はシート「成績処理2」のレポート提出の欄に「○」印を付けるプログラムを作成しなさい。
    Sheets("レポート").Select
    Range("A65536").End(xlUp).Select
    report_last_row = ActiveCell.Row

    Sheets("成績処理2").Select
    Range("A65536").End(xlUp).Select
    seiseki_last_row = ActiveCell.Row
    
    For i = 2 To report_last_row
        bangou = Sheets("レポート").Range("A" & i)
        For j = 2 To seiseki_last_row
            If bangou = Range("A" & j) Then
                Range("C" & j) = "○"
            End If
        Next j
    Next i
（注意！）
　複数のシートを操作するプログラムは「標準モジュール」の中に記述する必要があります。ワークシートの中に記述するとエラーが発生します。

＜課題5＞
　シート「レポート」に書いてある名前コードが「成績処理2」の中に存在しない場合、「名前コードxxxxが名簿にありません」とイミディエイトウィンドウに表示するよう、プログラムを改良しなさい。
[bookmark: _GoBack]　プログラムが正しく動作することを確認するため、存在しない名前コードを、シート「レポート」に書き加えて実行し、確認して下さい。
（ヒント）
　名前コードbangouを名簿の中で見つけたか否かを保持する変数を1つ導入します。変数名をfindとするなら、探索開始前にfind = 0とし、見つけたらfind = 1とします。検索終了後、find = 0なら、「名簿にない」というエラーを表示します。


＜課題6＞
　シート「成績処理2」「テスト」を使います。シート「テスト」にテストを受けた者の名前コードと点数のリストがあります。欠席者の行はありません。シート「成績処理2」のテスト点の欄を埋めなさい。

＜例題5＞
　シート「成績処理2」を使います。レポートを提出していない者のレポート欄に「×」を書き込み、その人の名前を薄い黄色RGB(255,255,204) で塗りつぶすプログラムを書きなさい。
    Sheets("成績処理2").Select
    Range("A65536").End(xlUp).Select
    seiseki_last_row = ActiveCell.Row
    
    For i = 2 To seiseki_last_row
        If Range("C" & i) = "" Then
            Range("C" & i) = "×"
            Range("B" & i).Interior.Color = RGB(255, 255, 204)
        End If
    Next i
＜課題7＞
　シート「成績処理2」を使います。テスト欠席者の欄に「欠」を書き込みなさい。また、テスト点が60点未満の点数欄を薄いピンクRGB(255,204,255) で塗りつぶすプログラムを書きなさい。

＜課題8＞
　ファイルsheet_manage2.xlsmを使います。2～4枚目のシートに1組～3組の生徒の名簿番号、名前、点数があります。シート "1組"を選ぶのは
    Sheets(2).Select              ' 2 枚目のシート
あるいは
    i = 1
    class = CStr(i) + "組"
    Sheets(class).Select
です。
　点数が60点未満の人は補習対象者です。補習対象者の点数欄をピンク色RGB(255,204,255) に着色し、シート「補習対象者」のシートに補習者の名簿を作成しなさい。
知っていると便利な関数
VBAが用意している関数
    a = Int(b)            ' 小数点以下切り捨て
    a = Round(b,2)        ' 四捨五入して小数点以下第2位まで求める
    a = Round(b,0)        ' 四捨五入して整数
    a = Sqr(b)            ' 平方根
    a = Abs(b)            ' 絶対値
    a = b Mod c           ' b ÷ c の余り
    a = b ^ c             ' 関数ではないが、べき乗の求め方
Excelが用意している関数
　「合計を求める(Sum)」「平均を求める(Average)」「○○の個数をカウントする(CountIf)」などの、Excelが持っている関数を使うことができます。

＜例題6＞
　シート「関数練習」を使います。セルA1～A10の合計を計算してA11に入れなさい。
    Range("A11") = WorksheetFunction.Sum(Range("A1:A10"))
　セルの範囲はA1:A10ではなく、Range("A1:A10") のように表します。この例では、A11に数値が入ります。A11に数式を入れる方法もあります。セルに数式を入れるには以下のように書きます。
    Range("A11").Formula = "=Sum(A1:A10)"
＜例題7＞
　セルB1～B10のうち、文字 "a" が入っているセルの個数をカウントしてB11に記入しなさい。
    Range("B11") = WorksheetFunction.CountIf(Range("B1:B10"), "a")
             ' 数値を書き込む
    Range("B11").Formula = "=CountIf(B1:B10,""a"")"
             ' 数式を書き込む
　上の例から分かるように、文字列の中に " が含まれる場合は、"" のように書きます。

＜例題8＞
　セルC1～C20に1～100までの乱数を書き込むプログラムを作成しなさい。
    For i = 1 To 20
        Range("A" & i) = WorksheetFunction.RandBetween(1, 100)
    Next i
＜例題9＞
　セルD1～D20に1～20までの乱数を書き込みなさい。ただし、1つの数は1回しか現れてはいけません。
　最初にD1～D20に1～20の数値を書き入れ、次に、20個のセルの数値をシャッフルします。n個の要素をシャッフルするアルゴリズムとして、次の2つが考えられます。

1. 「n個の要素から2つを選び、入れ替える」という操作を何回か繰り返す。具体的には以下のようにプログラムする：
1～nまでの2つの乱数a, bを生成し、a番目とb番目の要素を入れ替える。これを何回か繰り返す。
2. 「n個の要素の中から1個を取り出す。次に残りのn1個の要素の中から1個を取り出す。これを繰り返して取り出した要素を並べる」。具体的には以下のようにプログラムする：
1～nまでの乱数aを生成し、a番目の要素と最後の要素を入れ替える。最後の要素は確定。次に、1～n1番目の要素に対して同様の操作を行う。以下、操作の対象とする要素を1ずつ減らしながら繰り返す。

　プログラムしやすいのは1. の方法ですが、この方法は入れ替える作業を何回繰り返せばよいのかが明確ではなく、シャッフルした結果に偏りが生じるので、良くないそうです。シャッフルのアルゴリズムに関しては「配列　シャッフル　アルゴリズム」などで検索すると、色々なページが見つかります。以下のサイトが分かりやすいです。

http://www016.upp.so-net.ne.jp/garger-studio/gameprog/vb0311.html
http://ppp-lab.sakura.ne.jp/ProgrammingPlacePlus/algorithm/other/002.html

　1.の方法で組んだプログラムが以下です。30回という回数は適当です。
    For i = 1 To 20
        Range("D" & i) = i
    Next i
    
    For i = 1 To 30
        a = WorksheetFunction.RandBetween(1, 20)
        b = WorksheetFunction.RandBetween(1, 20)
        If a <> b Then
            tmp = Range("D" & a)
            Range("D" & a) = Range("D" & b)
            Range("D" & b) = tmp
        End If
    Next i
　2. の方法で組んだプログラムが以下です。
    For i = 1 To 20
        Range("D" & i) = i
    Next i
    
    For i = 20 To 2 Step -1
        a = WorksheetFunction.RandBetween(1, i)
        If a <> i Then
            tmp = Range("D" & i)
            Range("D" & i) = Range("D" & a)
            Range("D" & a) = tmp
        End If
    Next i
　ここで、新しい形式のForループが登場しました。
    For i = 20 To 2 Step -1
と書くと、iの値は20, 19, 18, 17, 16, ...... 4, 3, 2と変化します。For文の形式を一般化すると
    For 変数 = 初期値 To 終了値 Step 変化幅
となります。Step以下を省略すると1とみなされます。以下のような使い方もあります。
   For i = 1 To 10 Step 2
　この場合、i の値は1, 3, 5, 7, 9と2刻みに変化します。

＜課題9＞
　シート「ペア作成」を使います。以下の操作をするプログラムを作成しなさい。

1. B列をC列にコピーする。どのような命令を使うかは「開発」→「コード：マクロの記録」で見つけなさい。
2. C列をランダムに並べ直す。前の例題で1. の方法は良くないと書きましたが、2. の方法のプログラムを書くのは若干難易度が高いので、1. の方法で組んで構いません。
3. A列とC列のセルの値を全角空白を1個挟んで接続し、D列に書き込む。文字列が格納された変数aとbを全角空白1個を挟んで接続するには、a & "　" & b と書きます。あるいは a + "　" + b でも同じです。

